```
C ================================================================
C
C    ABAQUS UMAT for Linear Elastic Isotropic Material
C
C    Code adapted from:
C       http://birch.seas.harvard.edu/files/Writing%20a%20UMAT.pdf
C
C    Subroutine written by:
C       Vitor Takashi Endo, Eng.
C       endo.takashi@gmail.com
C       GRANTE - Group of Mechanical Design and Analysis
C       EMC - Mechanical Engineering Department
C       UFSC - Federal University of Santa Catarina
C
C ================================================================
      SUBROUTINE UMAT(STRESS, STATEV, DDSDDE, SSE, SPD, SCD, RPL,
     1 DDSDDT, DRPLDE, DRPLDT, STRAN, DSTRAN, TIME, DTIME, TEMP, DTEMP,
     2 PREDEF, DPRED, CMNAME, NDI, NSHR, NTENS, NSTATV, PROPS, NPROPS,
     3 COORDS, DROT, PNEWDT, CELENT, DFGRD0, DFGRD1, NOEL, NPT, LAYER,
     4 KSPT, KSTEP, KINC)
      INCLUDE 'ABA_PARAM.INC'
      CHARACTER*80 CMNAME
      DIMENSION STRESS(NTENS), STATEV(NSTATV), DDSDDE(NTENS, NTENS),
     1 DDSDDT(NTENS), DRPLDE(NTENS), STRAN(NTENS), DSTRAN(NTENS),
     2 PREDEF(1), DPRED(1), PROPS(NPROPS), COORDS(3), DROT(3, 3),
     3 DFGRD0(3, 3), DFGRD1(3, 3), EELAS(6), EPLAS(6), FLOW(6)
C ================================================================
C    Elastic constants for User Material
C     Young Modulus
      EMOD=PROPS(1)
C     Poisson Ratio
      ENU=PROPS(2)
C ================================================================
C    ELASTIC STIFFNESS MATRIX (in terms of E, v)
C----------------------------------------------------------------------------------------------------------
C [E(1-v)]/[(1+v)(1-2v)] [Ev]/[(1+v)(1-2v)]    [Ev]/[(1+v)(1-2v)]       0          0          0
C   [Ev]/[(1+v)(1-2v)]  E(1-v)]/[(1+v)(1-2v)]  [Ev]/[(1+v)(1-2v)]       0          0          0
C   [Ev]/[(1+v)(1-2v)]   [Ev]/[(1+v)(1-2v)]  [E(1-v)]/[(1+v)(1-2v)]     0          0          0
C         0                    0                    0           [E]/[2(1+v)]    0          0
C         0                    0                    0                0    [E]/[2(1+v)]    0
C         0                    0                    0                0          0   [E]/[2(1+v)]
C----------------------------------------------------------------------------------------------------------
C   Hooke's Law in Stiffness Form:
C       http://www.efunda.com/formulae/solid_mechanics/mat_mechanics/hooke_isotropic.cfm
      DO K1=1,NDI
        DO K2=1,NDI
          DDSDDE(K2,K1)=(EMOD*(ENU))/((1+ENU)*(1-2*ENU))
        END DO
        DDSDDE(K1,K1)=(EMOD*(1-ENU))/((1+ENU)*(1-2*ENU))
      END DO
      DO K1=NDI+1,NTENS
        DDSDDE(K1,K1)=(EMOD)/(2*(1+ENU))
      END DO
C ================================================================
C    STRESS CALCULATION
C----------------------------------------------------------------------------------------------------------
C [Sig11]   [ E(1-v)/(1+v)(1-2v) Ev/(1+v)(1-2v)    Ev/(1+v)(1-2v)       0      0       0     ] [eps11]
C [Sig22]   [  Ev/(1+v)(1-2v)  E(1-v)/(1+v)(1-2v)  Ev/(1+v)(1-2v)       0      0       0     ] [eps22]
C [Sig33] = [  Ev/(1+v)(1-2v)    Ev/(1+v)(1-2v)   E(1-v)/(1+v)(1-2v)    0      0       0     ] [eps33]
C [Tau12]   [         0              0                  0           E/2(1+v)  0       0     ] [gam12]
C [Tau13]   [         0              0                  0              0   E/2(1+v)   0     ] [gam13]
C [Tau23]   [         0              0                  0              0      0    E/2(1+v) ] [gam23]
C----------------------------------------------------------------------------------------------------------
      DO K1=1,NTENS
        DO K2=1,NTENS
          STRESS(K2)=STRESS(K2)+DDSDDE(K2,K1)*DSTRAN(K1)
        END DO
      END DO
C ================================================================
      RETURN
      END
```